#region license
// Copyright (c) 2007 Rodrigo B. de Oliveira (rbo@acm.org)
// All rights reserved.
// 
// Redistribution and use in source and binary forms, with or without modification,
// are permitted provided that the following conditions are met:
// 
//     * Redistributions of source code must retain the above copyright notice,
//     this list of conditions and the following disclaimer.
//     * Redistributions in binary form must reproduce the above copyright notice,
//     this list of conditions and the following disclaimer in the documentation
//     and/or other materials provided with the distribution.
//     * Neither the name of Rodrigo B. de Oliveira nor the names of its
//     contributors may be used to endorse or promote products derived from this
//     software without specific prior written permission.
// 
// THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
// ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
// WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
// DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE LIABLE
// FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL
// DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR
// SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER
// CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY,
// OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF
// THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

# Taken and modified from ConcurrentBoo

# This file is part of Grawlix.
# 
# Grawlix is free software: you can redistribute it and/or modify
# it under the terms of the GNU General Public License as
# published by the Free Software Foundation, either version 3 of
# the License, or (at your option) any later version.
# 
# Grawlix is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
# GNU General Public License for more details.
# 
# You should have received a copy of the GNU General Public
# License along with Grawlix.  If not, see
# <http://www.gnu.org/licenses/>.
#endregion

"""An unordered queue for asynchronous value computation."""

namespace ConcurrentBoo
import ConcurrentBoo.Primitive
import ConcurrentBoo.Executors
import System

transient class ComputationChannel:
"""Compute values asynchronously, saving the results."""
	
	def constructor():
	"""Default constructor, using a TaskPoolExecutor."""
		self(ThreadPoolExecutor())
	
	def constructor([required] executor as IExecutor):
	"""
	Construct a new channel with specific executor.
	Param executor: The IExecutor to use.
	"""
		_executor = executor
	
	def Enqueue([required] expression as Expression):
	"""
	Asynchronously sends an expression for evaluation.
	The completed computation can be received with Dequeue.
	Param expression: The expression to invoke.
	Returns: The computation, if wanted.
	"""
		return Enqueue(Computation(), expression)
	
	def Enqueue([required] computation as Computation,
	            [required] expression as Expression):
	"""
	Asynchronously sends an expression for evaluation.
	The completed computation can be received with Dequeue.
	Param computation: The computation to store the result.
	Param expression: The expression to invoke.
	Returns: The computation, if wanted.
	"""
		_executor.Execute() do:
			_channel.Enqueue(Invoke(computation, expression))
		return computation
	
	def Dequeue():
	"""
	Get the next result from the channel, waiting indefinitely.
	Returns: A completed Computation.
	"""
		return _channel.Dequeue()
	
	def TryDequeue(ref data as Computation):
	"""
	Get the next value from the channel, without waiting.
	Param data: out: The value found.
	Returns: true if take was successful, false on time out.
	"""
		return _channel.TryDequeue(data)
	
	def TryDequeue(timeout as timespan, ref data as Computation):
	"""
	Get the next computation from the channel, waiting as specified.
	Param timeout: Amount of time to wait.
	Param data: out: The completed computation found.
	Returns: true if take was successful, false on time out.
	"""
		return _channel.TryDequeue(timeout, data)
		
	def TryDequeue(millisecondsTimeout as int, ref data as Computation):
	"""
	Get the next computation from the channel, waiting as specified.
	Param millisecondsTimeout: Amount of time to wait, in milliseconds.
	Param data: out: The completed computation found.
	Returns: true if take was successful, false on time out.
	"""
		return _channel.TryDequeue(millisecondsTimeout, data)
	
	private final _executor as IExecutor
	"""The queue executor."""
	
	private final _channel = Channel[of Computation]()
	"""The internal queue."""
	
